
Capítulo 11
Transmissão do arquivo de pedidos por FTP

Neste capítulo, vamos utilizar as funções de API para o serviço de FTP, que permitirá 

o acesso a uma máquina remota para enviar arquivos.

FTP – File Transfer Protocol
FTP (File Transfer Protocol – Protocolo de Transferência de Arquivos) é um serviço 

para, basicamente, enviar e receber arquivos de um computador para outro na 

Internet. 

Para realizar essa implementação, adicione um novo formulário ao projeto. Inclua 

os seguintes componentes: 4 TextBox, 3 Label e 3 CommandButton. Salve o projeto 

e este novo formulário receberá o nome de frmFTP e será semelhante ao da figura 

11.1. As propriedades dos componentes estão nas tabelas 11.1 a 11.11.

Figura 11.1 – Formulário para Transmissão via FTP.

Tabela 11.1 – Propriedades do componente frmFTP

Propriedade Atributo
(Name) frmFTP 

Caption Carga

ScaleHeight 3960

ScaleWidth 3480

Tabela 11.2 – Propriedades do componente lblHost

Propriedade Atributo
(Name) lblHost 

AutoSize True

Caption Host (servidor FTP)

Height 195

Left 45

Top 0

Width 1425

Tabela 11.3 – Propriedades do componente txtHost

Propriedade Atributo
(Name) txtHost 

Height 255

Left 45

Locked False

MaxLength 0

Text

Top 240

Width 3255

Tabela 11.4 – Propriedades do componente lblOrigem

Propriedade Atributo
(Name) lblOrigem 

AutoSize True

Caption Localização Arq. Ori-
gem

Height 195

Left 45

Top 600

Width 1755

Tabela 11.5 – Propriedades do componente txtOrigem

Propriedade Atributo
(Name) txtOrigem 
Height 255
Left 45
Locked False
MaxLength 0
Text
Top 840
Width 3255

Tabela 11.6 – Propriedades do componente lblDestino

Propriedade Atributo
(Name) lblDestino 
AutoSize True
Caption Localização Arq. Destino
Enabled True
Height 195
Left 45
Top 1200
Width 1785

Tabela 11.7 – Propriedades do componente txtDestino

Propriedade Atributo
(Name) txtDestino 
Height 255
Left 45
Locked False
MaxLength 0
Text
Top 1440
Width 3255

Tabela 11.8 – Propriedades do componente txtMsg

Propriedade Atributo
(Name) txtMsg 
Height 1455
Left 45
Locked False
MaxLength 0
Text
Top 1800
Width 3255

Tabela 11.9 – Propriedades do componente cmdVoltar

Propriedade Atributo
(Name) cmdVoltar 

Caption Voltar

Height 375

Left 45

Top 3360

Width 975

Tabela 11.10 – Propriedades do componente cmdReceber

Propriedade Atributo
(Name) cmdReceber 

Caption Receber

Height 375

Left 1080

Top 3360

Width 1095

Tabela 11.11 – Propriedades do componente cmdEnviar

Propriedade Atributo
(Name) cmdEnviar 

Caption Enviar

Height 375

Left 2280

Top 3360

Width 975

Utilizaremos o código que está descrito na página da Microsoft, no endereço 

http://support.Microsoft.com/default.aspx?scid=kb;en-us;Q305598, e que foi adaptado para este 

projeto. Para que você possa realizar os testes de transferência, é necessário ter 

acesso a um servidor FTP. Para isso, é fundamental saber o nome do servidor ou 

seu endereço IP e ter privilégios para acessar os diretórios do servidor e realizar 

algumas operações (ler, gravar, excluir etc.).

O primeiro código deverá ser colocado no evento Click do botão de comando 

Voltar; para isso, dê dois cliques no cmdVoltar e inclua o código a seguir:

Private Sub cmdVoltar_Click()
 frmFTP.Hide
 frmTela_Cliente.Show
End Sub

No código anterior, quando o usuário clicar no botão Voltar, irá gerar o evento 

Click e, nesse evento, o formulário frmFTP será ocultado por meio do método Hide. 

O formulário frmTela_Cliente será exibido por meio do método Show. 

Dê dois cliques em qualquer área livre do formulário frmFTP, selecione o evento 

Activate e inclua o código a seguir:

Private Sub Form_Activate()
 ‘redesenha o formulário, atualizando-o
 frmFTP.Refresh
 ‘habilita botão de comando
 frmFTP.cmdEnviar.Enabled = True
 ‘define o nome do servidor que receberá o arquivo
 frmFTP.txtHost.Text = “SeuServidorFTP”
 ‘define a localização do arquivo que será enviado
 frmFTP.txtOrigem.Text = “\SuaPastaLocal\SeuArquivoLocal.TXT”
 ‘define o arquivo que será copiado para o servidor FTP
 frmFTP.txtDestino.Text = “SuaPastaRemoto/SeuArquivoRemoto.TXT”
End Sub

No código anterior, você deverá substituir onde houver SeuServidorFTP pelo 

nome ou endereço IP de seu servidor FTP. SuaPastaLocal é o caminho onde está 

localizado o arquivo a ser enviado do Pocket PC ou Emulador. SeuArquivoLocal é o 

nome do arquivo que deverá ser enviado.

Em SuaPastaRemoto, você deverá substituir pelo caminho de seu servidor FTP; caso 

você não deseje utilizar nenhum caminho, basta retirar essa variável. SeuArquivoRemoto é o 

nome do arquivo que será recebido pelo servidor FTP. Se desejar, poderá alterar o nome 

desse arquivo ou manter o nome original do arquivo que será enviado (SeuArquivoLocal). 

Atente para a barra separadora do caminho e nome do arquivo “ / “.

Clique duas vezes no botão de comando cmdEnviar e inclua o código a seguir:

Private sub cmdEnviar_Click
 On Error Resume Next
 Dim blnBin As Boolean
 Dim strUsuario As String
 Dim strSenha As String
 
 blnBin = False
 frmFTP.cmdEnviar.Enabled = False
 strUsuario = ”SeuUsuario”
 strSenha = “SuaSenha”
 frmFTP.txtMsg.Text = “Inicializado FTP”
 
 If InitializeFTP Then
  frmFTP.txtMsg.Text = “FTP inicializado”
  frmFTP.txtMsg.Text = “Conectando ao servidor FTP”
  If ConnectToFTPServer(frmFTP.txtHost.Text, strUsuario, strSenha) Then
   frmFTP.txtMsg.Text = “Conectado ao servidor FTP”

   frmFTP.txtMsg.Text = “Transferindo arquivo “ & frmFTP.txtOrigem.Text
   PutFileOnFTPServer frmFTP.txtOrigem.Text, _
    frmFTP.txtDestino.Text, blnBin
   frmFTP.txtMsg.Text = “Arquivo Transferido”
  End If
  CloseFTP
  frmFTP.txtMsg.Text = “Conexão Fechada”
 End If
 
 If Err.Number <> 0 Then
  txtMsg.Text = “Erro em cmdEnviar1_Click: “ & vbCrLf & _
      “Cód. Erro: “ & _
      Err.Number & vbCrLf & _
      “Descrição: “ & Err.Description
 End If
End Sub

No código anterior, se o servidor de FTP necessitar que sejam informados o usuá-

rio e a senha, substitua as informações nas variáveis strUsuario e strSenha, caso não 

necessite, apenas apague o conteúdo das variáveis, mas deixe-as entre aspas. 

O procedimento para realizar o FTP segue estes passos: primeiramente, devemos 

inicializar o serviço de FTP; fazemos isso por meio da função InitializeFTP que 

retornará True (verdade), se a inicialização ocorreu com sucesso, ou False (falso), 

caso tenha havido algum problema; neste caso, será apresentada uma mensagem 

reportando o erro.

Caso o serviço tenha sido inicializado com sucesso, iremos obter o handle deste 

servidor (handle é um ponteiro, ou seja, uma variável que contém o endereço de 

outra variável que é armazenado em memória, em um lugar fixo). Agora, é necessária 

a conexão com o servidor de FTP. Para isso, chamamos a função ConnectFTPServer, 

informando-a dos seguintes parâmetros: nome do servidor, usuário e senha. 

Caso a conexão tenha ocorrido com sucesso, serão retornados True e o handle 

desta conexão. A partir dessa conexão, poderemos realizar qualquer operação no 

servidor, desde que tenhamos os devidos privilégios para efetuar tais tarefas. No 

caso de dúvida, fale com o administrador de sua rede.

Os dois principais comandos para transmissão de arquivos são Get e Put. Com o 

Get, você faz o download de arquivos, isto é, traz do servidor FTP para o Pocket PC. 

Com o Put, você faz o inverso, envia o arquivo do Pocket PC para o servidor FTP. 

Como desejamos enviar o arquivo, iremos utilizar a função PutFileOnFTPServer, 

onde passamos os seguintes parâmetros: o arquivo origem, o arquivo destino e o 

tipo de transferência de arquivo, que será o formato ASCII ou Binary. 

Os códigos 11.1 a 11.6 devem ser incluídos no formulário frmFTP, já que serão 

utilizados apenas nesse formulário. Basicamente com esses códigos, você será capaz 

de enviar arquivos via FTP sem problemas.

Observe que o código 11.1 vem logo após a linha Option Explicit. O código 11.6 

é a função com os erros mais comuns que podem ocorrer neste tipo de serviço.

 Código 11.1 – Código do evento Click do cmdEnviar
Const INTERNET_OPEN_TYPE_PRECONFIG = 0
Const INTERNET_OPEN_TYPE_DIRECT = 1
Const INTERNET_OPEN_TYPE_PROXY = 3
Const INTERNET_FLAG_RELOAD = &H80000000
Const FTP_TRANSFER_TYPE_UNKNOWN As Long = 0
Const FTP_TRANSFER_TYPE_ASCII As Long = 1
Const FTP_TRANSFER_TYPE_BINARY As Long = 2
Const INTERNET_DEFAULT_FTP_PORT As Long = 21
Const INTERNET_FLAG_PASSIVE As Long = &H8000000
Const INTERNET_SERVICE_FTP As Long = 1
Declare Function FtpPutFile Lib “wininet” Alias “FtpPutFileW” ( _
 ByVal hFtp As Long, _
 ByVal lpszLocalFile As String, _
 ByVal lpszNewRemoteFile As String, _
 ByVal dwFlags As Long, ByVal dwContext As Long) As Long
Declare Function InternetCloseHandle Lib “wininet” _
 (ByVal hInet As Long) As Long
Declare Function InternetConnect Lib “wininet” Alias “InternetConnectW” ( _
 ByVal hInet As Long, _
 ByVal lpszServerName As String, _
 ByVal nServerPort As Long, _
 ByVal lpszUsername As String, _
 ByVal lpszPassword As String, _
 ByVal dwService As Long, _
 ByVal dwFlags As Long, _
 ByVal dwContext As Long) As Long
Declare Function InternetOpen Lib “wininet” Alias “InternetOpenW” ( _
 ByVal lpszAgent As String, _
 ByVal dwAccessType As Long, _
 ByVal lpszProxyName As String, _
 ByVal lpszProxyBypass As String, _
 ByVal dwFlags As Long) As Long
Declare Function GetLastError Lib “coredll” () As Long
Public lngInternetHandle As Long
Public lngFtpHandle As Long
Public hOpenUrl As Long

 Código 11.2 – Sub-rotina para inicializar serviço de FTP
Public Function InitializeFTP() As Boolean
 ‘inicialização do serviço de FTP
 On Error Resume Next
 Dim sUrl As String

 lngInternetHandle = InternetOpen(“eVB OpenUrl”, _
  NTERNET_OPEN_TYPE_PRECONFIG, _
  vbNullString, vbNullString, 0)
 If Err.Number <> 0 Or lngInternetHandle = 0 Then
  MsgBox “Falha para inicializar a comunicação” & vbCrLf & _
    “para transferir dados.” & vbCrLf & _
    Err.Description, vbInformation + vbOKOnly, _
    “ERRO”
  InitializeFTP = False
 Else
  InitializeFTP = True
 End If
End Function

 Código 11.3 – Sub-rotina para conectar com servidor de FTP
Public Function ConnectToFTPServer( _
 strFTPServerName As String, _
 strUsername As String, _
 strPassword As String) As Boolean
 ‘conectar com servidor de FTP, informando o nome do servidor, usuário e senha
 On Error Resume Next
 lngFtpHandle = InternetConnect( _
  lngInternetHandle, _
  strFTPServerName, _
  INTERNET_DEFAULT_FTP_PORT, _
  strUsername, strPassword, _
  INTERNET_SERVICE_FTP, _
  INTERNET_FLAG_PASSIVE, 0)
 If lngFtpHandle = 0 Or Err.Number <> 0 Then
  MsgBox “Falha para conectar servidor” & vbCrLf & _
     “e transferir dados” & vbCrLf & _
     “Cód. do Erro: “ & vbCrLf & _
     CStr(GetLastError), vbInformation + _
     vbOKOnly, “ERRO”
  MsgBox GetErr(GetLastError), vbInformation + vbOKOnly, “ERRO”
  ConnectToFTPServer = False
 Else
  ConnectToFTPServer = True
 End If
End Function

 Código 11.4 – Sub-rotina para upload de arquivo
Public Function PutFileOnFTPServer( _
 ByVal strSourceFilename As String, _
 ByVal strDestFilename As String, _
 ByVal blnBinaryFile As String) As Boolean
 ‘transfere o arquivo local para servidor
 On Error Resume Next
 Dim blnTransferredOK As Boolean
 Dim lngTransferType As Long
 If blnBinaryFile Then
  lngTransferType = FTP_TRANSFER_TYPE_BINARY

 Else
  lngTransferType = FTP_TRANSFER_TYPE_ASCII
 End If
 blnTransferredOK = FtpPutFile(lngFtpHandle, strSourceFilename, _
          strDestFilename, lngTransferType, 0)
 If blnTransferredOK = 0 Then
  MsgBox “Transferência do arq: “ & vbCrLf & _
    strSourceFilename & vbCrLf & _
    “para o servidor falhou.” & vbCrLf & _
    “Cód. Erro: “ & CStr(GetLastError()), vbInformation + _
    vbOKOnly, “ERRO”
 Else
  MsgBox “Sucesso na Transferência do arq.:” & vbCrLf & _
   strSourceFilename, vbInformation + vbOKOnly, “ERRO”
 End If
 PutFileOnFTPServer = blnTransferredOK
End Function

 Código 11.5 – Sub-rotina para encerramento da conexão e serviço de FTP
Public Sub CloseFTP()
 On Error Resume Next
 InternetCloseHandle lngFtpHandle
 InternetCloseHandle lngInternetHandle
End Sub

 Código 11.6 – Sub-rotina que retorna mensagem de erro
Public Function GetErr(lErrorCode As String) As String
 Dim sBuffer As String
 Dim nBuffer As Long
 lErrorCode = CLng(lErrorCode)
 Select Case lErrorCode
  Case “12002”
   GetErr = “Excedeu o Limite de Tempo.Tente mais tarde”
  Case “12004”
   GetErr = “Verifique se está tudo bem encaixado e “ & _
      “conectado.Tente novamente”
  Case “12007”
   GetErr = “Nome do Servidor não resolvido. Tente mais tarde”
  Case “12008”
   GetErr = “Protocolo FTP não encontrado. Tente mais tarde”
  Case “12013”
   GetErr = “Nome de Usuário inválido. Tente mais tarde”
  Case “12014”
   GetErr = “Senha inválida. Tente mais tarde”
  Case “12015”
   GetErr = “Falha no Login. Tente mais tarde”
  Case “12110”
   GetErr = “Caiu Conexão. Tente mais tarde”

  Case Else
   GetErr = “Verifique se está tudo bem encaixado e “ & _
      “conectado.Tente novamente”
 End Select
End Function

Falta ainda incluirmos a chamada do formulário frmFTP. Para isso, selecione o formu-

lário frmTela_Transmissao, dê dois cliques no botão cmdFTP e inclua o código a seguir:

Private Sub cmdFTP_Click()
 frmFTP.Show
End Sub

Existem alguns detalhes a serem observados em relação ao FTP. Se você estiver 

utilizando um Pocket PC com o PPC 2002, existe um bug nesta versão, no qual o 

serviço de FTP não reconhece o usuário e a senha informada em sua aplicação; 

somente irá logar-se como usuário anônimo. 

Se você estiver utilizando o iPAQ, da Compaq, com PPC 2002, deverá baixar o 

Service Pack iPAQ Pocket PC 2002 End User Update (EUU – Service Pack). Em http://

h18000.www1.hp.com/support/files/handhelds/us/download/15068.html, selecione o idioma desejado 

e siga as instruções contidas no site para realizar esta atualização.

Para os demais fabricantes, é necessário contatá-los e verificar se existe algum tipo 

de atualização, como ocorre com a HP, ou tentar algo com o suporte da Microsoft.

No caso do Windows Mobile 2003, este código não será executado com sucesso, 

pois, por questão de segurança, a Microsoft não evoluiu este serviço em seu sistema 

operacional. Você somente conseguirá conectar-se como anônimo. Não existe ne-

nhuma atualização para essa versão, qualquer que seja o fabricante. A solução para 

utilizar o FTP no Windows Mobile 2003 é recorrer a componentes pagos. A seguir, 

estão listados alguns links onde você poderá baixar uma versão de avaliação para 

testes e, se gostar, poderá efetivar a compra no mesmo site.

http://www.sapphire-solutions.co.uk/products/ftp/

http://www.triliumsoftware.com/evb/index.html

http://www.softempire.com/scotty-ftp-v1-17p-windows-ce-2-0.html

Softwares de FTP que não sejam ActiveX podem ser encontrados no endereço 

http://www.pocketgear.com/products_search.asp?searchfor=FTP&x=39&y=2.

Até o momento em que este livro foi elaborado, todos os links informados esta-

vam ativos, entretanto, se porventura algum deles retornar como não encontrado, 

tente fazer uma pesquisa no Google (http://www.google.com.br).


